

SysplorerByExample

汇报人：王慧敏 孔德才



课程须知

本课程基于单机版本：Sysplorer 2025b构建

本课程面向Modelica的语言的基础学习，学习本课程前须熟悉Sysplorer基本功能

课程基于案例和Modelica语法进行回顾

具体语法的详解参见Sysplorer基本课程-《Modelica语法》

@苏州同元软控信息技术有限公司版权所有，未经许可，不得复印、传播或以其他方式使用

@苏州同元软控信息技术有限公司
版权所有，侵权必究

目录



复习与简单案例



一维热传导



考虑摩擦副的滑块运动建模



管道建模

主要的声明前缀

声明前缀 (Declaration Prefixes)，是一组用于修饰变量声明的关键字。它们为变量赋予了特定的属性和行为，极大地增强了 Modelica 语言的表达能力和灵活性。理解这些前缀对于编写高质量、可重用和结构清晰的 Modelica 模型至关重要

前缀	类别	主要作用	示例
parameter	可变性	仿真前可设置、仿真中不变的参数	parameter Real m=1;
constant	可变性	编译时固定、永不改变的常量	constant Real g=9.81;
input	因果关系	外部提供的输入信号	input Real u;
output	因果关系	模型计算产生的输出信号	output Real y;
discrete	可变性	值仅在离散时间点变化的变量	discrete Integer counter;
flow	因果关系	连接器中	flow Real i;
protected	可见性	隐藏内部实现，外部不可访问	protected Real internalState;
inner/outer	结构	跨层级传播环境或全局属性	outer Environment env;
replaceable	结构	允许组件/模型被子类型替换	replaceable model M = BaseM;
final	结构	禁止后续修改或重新声明	final parameter Real p=10;

常用的类

Modelica中的类（Class）是建模的基本单元，用于描述物理组件的属性、行为和接口。

核心特性：

- 支持面向对象设计（继承、多态、封装）
- 基于方程（非因果建模）
- 支持多种类类型（model, connector, record等）

类型	名称	作用
model	模型	定义组件模型
connector	连接器	定义组件之间的连接接口
record	记录	数据结构
block	框图	兼容基于框图的因果建模
type	类型	类型别名
function	函数	通过算法实现过程式建模
package	包	消除名字冲突和组织模型层次，本质上相当于一个文件夹

简单的一阶系统-入门

首先，让我们考虑一个非常简单的微分方程: $\frac{dx}{dt} = 1 - x$

在这个方程中，只有一个变量x。此方程可以用Modelica 语言表述如下：

```
model FirstOrder "简单一阶"  
  Real x "变量x";  
  equation  
    der(x) = 1 - x;  
end FirstOrder;
```

这段代码以关键词model 开始，以表明模型定义的开始。后面紧跟着模型的名字FirstOrder，再后面是变量声明部分。

从这个方程中，我们可以确定变量x 是一个连续实数。在Modelica 语言中，我们可以用Real x; 语句来声明它。Real 类型只是众多数据类型的一种（具体关于数据类型请参见Modelica语法课程）。


当所有的变量声明完成后，我们就可以编写描述模型特性的方程了。在这个例子中，我们可以用运算符der 来表示变量x 的一阶导数

接下来：涉及求解过程的繁琐工作都是由Modelica 编译器自动处理

$$\text{der}(x) = 1 - x$$

简单的一阶系统-入门



$$\frac{dx}{dt} = 1 - x \quad \xrightarrow{\text{通过分离变量并积分，得到了一阶常微分方程的通解}} \quad x(t) = 1 - Ce^{-t} \quad \text{其中} C \text{为任意常数}$$



在微分方程求解中，初始值是指求解微分方程时，在初始时刻对未知函数及其导数所给定的值。初始值对微分方程的求解至关重要，因为它决定了微分方程解的唯一性、存在性以及求解结果的准确性。

在Sysplorer求解算法中，若没有给定变量的初值，则会默认变量初值为0


所以想要保证结果准确性，需要给定合适的初值



简单的一阶系统-入门

我们采用了更简洁的方法。我们可以在声明变量的start 属性时直接指定其初始条件。其实，start 属性更多的是一种绑定关系的暗示。如果Modelica 编译器将某个特定变量识别为状态（即某个变量需要一个初始化条件），并且模型的initial equation 部分没有提供非充足的初始条件，这时start 属性就可以作为替代，声明变量的初始条件。换句话说，你可以认为start 属性是在必须具备初始条件的情况下的“后备初始条件”。可以使用fixed 属性。变量的fixed 属性被用来通知编译器start 属性必须作为初始条件来使用。换句话说，如下的initial equation 等同于start =xxx, fixed=true

```
model Firstorder "简单一阶"  
  Real x(start=2)"变量x";  
  initial equation  
  x=3;          初始方程  
  equation  
  der(x) = 1-x;  
end Firstorder;
```



```
model Firstorder "简单一阶"  
  Real x(start=2,fixed=true)"变量x";  
  initial equation  
  x=3;  
  equation  
  der(x) = 1-x;  
end Firstorder;
```

正在分析模型...

错误(6142): 模型中 Real 类型的初始条件有冗余，如下集合中有 1 个多余初始条件：

```
x = 2;  
x = 3;
```

---- 翻译失败终止 ----

编译模型"Firstorder"失败。

简单的一阶系统-入门


文档而非注释：

注意在这个模型中引号中的文字。

理解这些引号中的文字模块非常重要，它们在计算机科学中被称为“字符串”，而不是注释（如同代码中的//测试注释的文本）。这些“描述性字符串”不像注释，不能在任意的地方插入。相反的，它们只能在特定的地方插入，以提供与其关联的模型元素的附加说明。


对模型做相应的注释，文档语句必须插入在模型名字的后面。

在后面我们将会看到，这种文档可以被很多工具所使用。例如，搜索模型时，搜索工具可以利用这些描述性的字符串进行识别匹配。这些文本也可以与模型的图示相关联。



下面这个公式是一个描述系统能量平衡的一阶常微分方程，通常用于分析容器或管道中流体的加热/冷却过程

$$\rho V C_p \frac{dT_o}{dt} = \dot{m}_i C_p (T_i - T_o) + hA(T_s - T_o)$$



$\rho V C_p \frac{dT_o}{dt}$ 表示系统内流体温度随时间变化的能量累积项

ρ : 流体密度 (kg / m^3)

V : 流体体积 (m^3)

C_p : 流体的定压比热容 ($\text{J} / \text{kg} \cdot \text{K}$)

$\frac{dT_o}{dt}$: 出口温度 T 对时间的导数 (K / s) 反映系统温度变化的快慢

$\dot{m}_i C_p (T_i - T_o)$ 对流项: 描述流体流入带来的能量交换

$\dot{m}_i$: 入口质量流量 kg / s

T_i : 入口流体温度 (K)

T_o : 出口流体温度 (K)

$\dot{m}_i C_p (T_i - T_o)$ 对流项: 描述流体流入带来的能量交换

$\dot{m}_i$: 入口质量流量 kg / s

T_i : 入口流体温度 (K)

T_o : 出口流体温度 (K)

$$\rho V C_p \frac{dT_o}{dt} = \dot{m}_i C_p (T_i - T_o) + hA(T_s - T_o)$$

```
model Easy_Heat "简单热交换"
parameter Real rou=1000;
parameter Real V=0.2;
parameter Real Cp=1.8;
parameter Real h=1000;
parameter Real A=250;
parameter Real T_in=80;
parameter Real T_s=150;
parameter Real m=20;
Real T_out(start=75);
equation
rou*V*Cp*der(T_out)=m*Cp*(T_in-T_out)+h*A*(T_s-T_out);
end Easy_Heat;
```

模型构建

左侧所示简单热交换的Modelica模型代码。

代码第一行使用关键字model声明了所建模型Easy_Heat；接下来声明了参量、变量。

参量是在一个仿真过程中保持特定值的量(如本例中的rou=1000)；

变量是时间的函数(如本例T_out)；


最后定义了模型包含的方程equation；代码以关键字end加模型名称结束。

在进行参量、常量(常数)和变量的声明时，都必须用到相应的关键字。如声明参量要用parameter关键字；对于变量要使用数据类型关键字声明，如T_out在本例中属于实型，要使用Real关键字声明。


仔细观察例中所定义的方程，不难发现示例包含了一个Modelica内置操作der，der是对其包含的变量求时间导数。

为参数与变量标示单位不仅可以提高代码可读性，且Modelica 编译器可以由此测试变量量纲是否匹配。出于这个原因，尽量在参数和变量里使用物理类型是相当有益的。

Modelica.SIunits 包是内容非常多，而且充满了很少使用的物理单位。之所以包含这些非常用单位是为了完整地遵守ISO 31-1992 标准。以下是例子说明了SIunits 包中常用物理单位是如何定义的：



我们可以使用type的类型，仿照Modelica.SIunits 包，自定义的单位，例如自定义对流换热系数，显示为w/k




物理建模

物理建模涉及到不同物理量(如密度、体积和质量等)之间的特定关系定义。

Modelica可以定义不同的物理量，并赋值给它们。为了说明Modelica 定义物理量的方法，这里还是我们这个模型。

```
model Easy_Heat "简单热交换"
parameter Modelica.SIunits.Density rou=1000;
parameter Modelica.SIunits.Volume V=2;
parameter Modelica.SIunits.SpecificHeatCapacity Cp=1.8;
parameter Modelica.SIunits.CoefficientOfHeatTransfer h=1000;
parameter Modelica.SIunits.Area A=250;
parameter Modelica.SIunits.Temp_C T_in=80;
parameter Modelica.SIunits.Temp_C T_s=150;
parameter Modelica.SIunits.MassFlowRate m=20;
Real T_out(start=75);
equation
rou*Cp*V*der(T_out)=m*Cp*(T_in-T_out)+h*A*(T_s-T_out);
end Easy_Heat;
```



定义在 Modelica.SIunits包中，所以在物理量类型定义时，包含Modelica.SIunits。如密度变量的定义，其格式为 **Modelica.SIunits.Density**。

01

PART 01 ➔

综合案例展示-一维热传导


@苏州同元软控信息技术有限公司版权所有，侵权必究

未经许可，不得复印、传播或以其他方式使用


1. 一维热传导

在这个例子中，我们将考虑如何用数组来表述更实际的问题，即变量的一维空间分布。我们将观察Modelica与数组有关的语言特性，以及这些特性如何帮助我们更简略地表达系统行为。


本例是关于一个简单热传导问题。考虑如下所示的一维棒：



一维棒示意图



热传导动画



截面积

第*i*节体积

第*i*个截面的热容 $m_i C T_i$

其中 $m_i = \rho V_i$ 为第*i*节的质量， C 为比热容， T_i 为第*i*节的温度。

第*i*节的体积 $V_i = A L_i$

其中 A 为导热棒的截面积。

任何时间第*i*节中的热量增长 = $\rho A L_i C \frac{dT_i}{dt}$

假定 A 、 L_i 和 C 不随时间变化。

1. 一维热传导

考虑两种不同形式的热传递：与环境的对流换热、传热棒内部相邻部分的热传导。

- 与环境的对流换热

$$q_h = -hA(T_i - T_{amb})$$

其中 h 为对流系数。

- 传热棒内部相邻部分的热传导其中之一是与 $i-1$ 元素的热传导（如果这个元素存在），另外则是与 $i+1$ 元素的热传导（前提也是该元素存在）。这些因素可以分别表示为：

$$q_{k \rightarrow i-1} = -kA \frac{T_i - T_{i-1}}{L_i}$$

$$q_{k \rightarrow i+1} = -kA \frac{T_i - T_{i+1}}{L_i}$$

- 利用所分析内容，可以得到首个元素的热平衡方程

$$\rho AL_i C \frac{dT_1}{dt} = -kA \frac{T_1 - T_2}{L_i} - hA(T_1 - T_{amb})$$

- 最后一个元素的热平衡方程

$$\rho AL_i C \frac{dT_n}{dt} = -kA \frac{T_n - T_{n-1}}{L_i} - hA(T_n - T_{amb})$$

- 其他元素的热平衡方程

$$\rho AL_i C \frac{dT_i}{dt} = -kA \frac{T_i - T_{i-1}}{L_i} - kA \frac{T_i - T_{i+1}}{L_i} - hA(T_i - T_{amb})$$

1. 一维热传导

参考代码

```
model Rod_ForLoop"使用 for 循环对杆内的热传导进行建模"
  annotation(__MWORKS(version="2025b"));
  import SI = Modelica.Units.SI;
  //定义单位
  type ConvectionCoefficient=Real(unit="W/K", min=0);
  type ConductionCoefficient=Real(unit="W.m-1.K-1", min=0);


  //定义常数
  constant Real pi = 3.14159;

  //定义参数
  parameter Integer n=10"对传热棒分段的段数";
  parameter SI.Length L=1.0"传热棒长度";
  parameter SI.Radius R=0.1 "传热棒半径";
  parameter SI.Density rho=2.0"材料密度";
  parameter ConvectionCoefficient h=2.0 "对流系数";
  parameter ConductionCoefficient k=10 "传导系数";
  parameter SI.SpecificHeatCapacity C=10.0 "比热";
  parameter SI.Temperature Tamb=300 "环境温度";
  parameter SI.Area A = pi*R^2 "传热棒截面积";
  parameter SI.Volume V = A*L/n "每一节传热棒体积";

  //定义变量
  SI.Temperature T[n];

  initial equation
    T = linspace(200,300,n);
  equation
    rho*V*C*der(T[1]) = -h*(T[1]-Tamb)-k*A*(T[1]-T[2])/(L/n);
    for i in 2:(n-1) loop
      rho*V*C*der(T[i]) = -k*A*(T[i]-T[i-1])/(L/n)-k*A*(T[i]-T[i+1])/(L/n) -h*(T[i]-Tamb);
    end for;
    rho*V*C*der(T[end]) = -h*(T[end]-Tamb)-k*A*(T[end]-T[end-1])/(L/n);
end Rod_ForLoop;
```

温度结果曲线



1. 一维热传导

可选其他方法

实际上，有好几种方法可以生成我们需要的方程。而每种方法在不同的情景下有其优点与缺点。而选择哪种方法可以让方程感觉上最容易理解，则是取决于模型开发者自己。

我们可以用一个数组特性让这些方程变得更为简单。这个特性叫做数组解析（译注：原文为array comprehension，类似Python的list comprehension）。数组解析将for循环倒了过来。也就是说，我们在单一的等式后加上了循环变量在循环时如何取不同值的信息。在我们的例子中，我们可以使用数组解析将方程表达为如下方式：

```
equation
rho*V*C*der(T[1]) = -h*(T[1]-Tamb)-k*A*(T[1]-T[2])/(L/n);
rho*V*C*der(T[2:n-1]) = {-k*A*(T[i]-T[i-1])/(L/n)-k*A*(T[i]-T[i+1])/(L/n)
for i in 2:(n-1)};
rho*V*C*der(T[end]) = -h*(T[end]-Tamb)-k*A*(T[end]-T[end-1])/(L/n);
```

Modelica语言支持矢量方程。在这些情况下，只要方程左右均是大小同样的向量，我们就可以使用一个（向量）方程来表示多个标量方程。我们可以利用这个特性来简化方程式，结果如下：

```
equation
rho*V*C*der(T[1]) = -h*(T[1]-Tamb)-k*A*(T[1]-T[2])/(L/n);
rho*V*C*der(T[2:n-1]) = -k*A*(T[2:n-1]-T[1:n-2])/(L/n)-k*A*(T[2:n-1]-
T[3:n])/(L/n);
rho*V*C*der(T[end]) = -h*(T[end]-Tamb)-k*A*(T[end]-T[end-1])/(L/n);
```

@苏州同元软控信息技术有限公司版权所有，侵权必究

1. 一维热传导

涉及知识点:

- Modelica基础语法:
 - import的使用;
 - type类的定义;
 - 数组的使用。
- Modelica简单函数:
 - 求导函数der();
 - for循环结构等。

@苏州同元软控信息技术有限公司版权所有，未经许可，不得复印、传播或以其他方式使用
版权所有，侵权必究

02

PART 02 ➔


案例展示-考虑摩擦副的滑块运动建模

@苏州同元软控信息技术有限公司
版权所有，侵权必究

未经许可，不得复印、传播或以其他方式使用

2.1 摩擦副建模-原理分析


不考虑摩擦副的滑块运动模型



测试工况:

组件参数			
常规			
参数			
offset	0		Offset of output signal y
startTime	0	s	Output y = offset for time < startTime
amplitude	10		Amplitude of sine wave
freqHz	1	Hz	Frequency of sine wave
phase	90	deg	Phase of sine wave

仿真结果曲线:



为了系统模型, 更贴近实际物理系统, 尝试开发一个带摩擦副的滑块运动模型

功能需求:

1. 能够描述摩擦力与物体表面接触压力之间的关系;
2. 能够描述不同条件下的摩擦力大小和方向, 包括干摩擦、粘性摩擦等情况;
3. 考虑动态摩擦和静态摩擦之间的转换, 以及滑动速度对摩擦力的影响。

摩擦副数学模型:

$$F_{stick} = \mu_s \cdot F_n$$

$$F_{slip} = \begin{cases} \mu_c \cdot F_n + (\mu_c \cdot F_n - F_{stick}) \cdot \exp\left(-3 \cdot \frac{|v|}{v_{strb}}\right) \cdot \text{sign}(v) & \text{考虑stirbeck效应} \\ \mu_c \cdot F_n \cdot \text{sign}(v) & \text{不考虑stirbeck效应} \end{cases}$$

$$F_{vis} = \mu_v \cdot v$$

$$F_{fri} = F_{slip} + F_{vis}$$

其中:

μ_s	静摩擦系数
μ_c	滑动摩擦系数
μ_v	粘性摩擦系数
v	速度
v_{strb}	Stribeck速度常数
F_n	法向力
F_{stick}	最大静摩擦力
F_{slip}	滑动摩擦力
F_{vis}	粘性摩擦力
F_{fri}	总摩擦力

2.2 摩擦副建模-参数变量和接口的定义

根据模型原理设置相关的参数和变量和接口

μ_s	静摩擦系数	参数
μ_c	滑动摩擦系数	参数
μ_v	粘性摩擦系数	参数
v	速度	变量
F_n	法向力	参数
v_{strb}	Stribeck速度常数	参数
F_{stick}	最大静摩擦力	变量
F_{slip}	滑动摩擦力	变量
F_{vis}	粘性摩擦力	变量
F_{fri}	总摩擦力	变量
$Flange_a$	力和位移	一维平动接口
$Flange_b$	力和位移	一维平动接口

```
model SlidingFriction "摩擦副"
//参数
parameter Boolean Stribeck = true "是否考虑Stribeck效应";
parameter Real mu_c = 0.1 "滑动摩擦系数";
parameter Real mu_s = 0.12 "静摩擦系数";
parameter Real mu_v(final unit = "N.s/m2") = 0.1 "粘性摩擦系数";
parameter Modelica.SIunits.Force fN = 10 "法向力";
parameter Modelica.SIunits.Velocity v_Strb = 0.01 "Stribeck速度常数";
//变量
Modelica.SIunits.Force f_fri "总摩擦力";
Modelica.SIunits.Force f_slip "滑动摩擦力";
Modelica.SIunits.Force f_stick "最大静摩擦力";
Modelica.SIunits.Force f_vis "粘性摩擦力";
Modelica.SIunits.Position s_rel "相对位移";
Modelica.SIunits.Velocity v_rel "相对速度";
//接口
Modelica.Mechanics.Translational.Interfaces.Flange_a
flange_a
  annotation (...);
Modelica.Mechanics.Translational.Interfaces.Flange_b
flange_b
  annotation (...);
end SlidingFriction
```


2.3 摩擦副建模-方程描述

根据模型原理公式编写方程代码

$$F_{stick} = \mu_s \cdot F_n$$

$$F_{slip} = \begin{cases} \left(\mu_c \cdot F_n + (\mu_c \cdot F_n - F_{stick}) \cdot \exp\left(-3 \cdot \frac{|v|}{v_{strb}}\right) \right) \cdot \text{sign}(v) & \text{考虑stribeck效应} \\ \mu_c \cdot F_n \cdot \text{sign}(v) & \text{不考虑stribeck效应} \end{cases}$$

$$F_{vis} = \mu_v \cdot v$$

$$F_{fri} = F_{slip} + F_{vis}$$

equation

```
s_rel = flange_b.s - flange_a.s;
```

```
v_rel = der(s_rel);
```

```
f_stick = mu_s * fN;
```

```
f_vis = mu_v * v_rel;
```

```
if Stribeck == true then
```

```
    f_slip = (mu_c * fN + (mu_c * fN - f_stick) * exp(-3 * abs(v_rel) /  
v_Strb)) * sign(v_rel);
```

```
else
```

```
    f_slip = mu_c * fN * sign(v_rel);
```

```
end if;
```

```
f_fri = f_slip + f_vis;
```

```
flange_b.f = f_fri;
```

```
flange_a.f = -f_fri;
```


陈述式建模

2.4 摩擦副建模-绘制图标

模型构造完成之后，绘制模型图标：

操作步骤：

- 选择图标界面
- 使用“建模”或“编辑”的基本图元进行绘制
- 在图形属性中，为图形添加填充颜色及样式
- 在图标界面通过“%name”添加模型名称



name



滑动摩擦副图标


注意：需要结合接口
绘制组件图标

@苏州同元软控信息技术
版权所有，侵权必究

2.5 摩擦副建模-单元测试

对构建的模型进行测试，查看仿真结果是否满足功能要求：



单元测试：



滑动摩擦副单元测试模型

测试工况：

组件参数			
常规			
参数			
offset			Offset of output signal y
startTime	0	s	Output y = offset for time < startTime
amplitude	10		Amplitude of sine wave
freqHz	1	Hz	Frequency of sine wave
phase	90	deg	Phase of sine wave



仿真结果曲线

结果分析：

- 相同力的驱动下，加了摩擦副的质量块运动距离明显缩短；
- 摩擦力的作用方向和运动方向相反，满足实际物理要求。

03



PART 03 ➔

案例展示-管道建模

@苏州同元软控信息技术
版权所有，侵权必究

版权所有，未经许可，不得复印、传播或以其他方式使用


管道模型




不得复印、传播或以其他方式使用

@苏州同元软控信息技术有限公司版权所有，侵权必究

2.架构实现




3.架构实现



@苏州同元软控信息技术有限公司版权所有，侵权必究

3.模型实现



接口代码实现

```
connector FluidInterface "流体接口"  
    Modelica.SIunits.Pressure p "压力";  
    flow Modelica.SIunits.MassFlowRate m_flow "流量";  
    stream Modelica.SIunits.SpecificEnthalpy h "比焓";  
end FluidInterface;
```

相关知识点

- 数据类型
- 连接器
- 流势变量

因果连接器	具有因果传递顺序的信号流连接器，一般用于控制系统建模
非因果连接器	根据广义基尔霍夫定律建立的能量流连接器，用于机电液热控等物理领域建模
可扩展连接器	可根据接入变量扩展接口变量的总线连接器，一般用于总线定义
隐式连接器	通过inner/outer配合使用的隐式连接器，一般用于读取环境变量和介质物性。

3.模型实现

数据类型：

- Real, Integer, Boolean, String, enumeration

连接器：


- 连接器就是连接器类（connector）的实例，用于定义组件与外部的通信接口

流势变量：

- 势变量（非流变量）： 没有flow前缀的变量，一般表示能量水平；
- 流变量（flow）： 有flow前缀的变量；

@苏州同元软控信息技术有限公司版权所有，侵权必究

3.模型实现



全局变量

```
model System "全局变量"
  parameter SI.Pressure p0 = 1.0135e5 "环境压力";
  parameter SI.Temperature T0 = 293.15 "环境温度";
  parameter SI.Acceleration g = Modelica.Constants.g_n "重力加速度";
  parameter SI.Density rho = 1000 "液体密度";
  parameter SI.DynamicViscosity mu = 0.001 "动力粘度";
  parameter SI.SpecificHeatCapacityAtConstantPressure cp = 4181 "定压比热容";
  parameter SI.ThermalConductivity lambda = 0.598 "导热系数";
  annotation (defaultComponentName = "system",
    defaultComponentPrefixes = "inner",
```

相关知识点:

- inner/outer;
- 类型前缀: parameter

3.模型实现


inner/outer

- 提供了一种外层变量或外层类型的引用机制;
- inner/outer 相当于定义了一个全局接口或变量，可以在嵌套的所有实例层次中被访问
- defaultComponentName
- defaultComponentPrefixes

parameter

- 参数面板输入，在单次仿真过程中不改变值;

3.模型实现



管道模型理论

管道压降:

$$P_{in} - P_{out} = j \frac{dm_{out}}{dt} + \frac{1}{\rho} \xi_f \dot{m}_{out}^2 + \rho g \Delta h$$

惯性损失:

$$j = l / A$$

沿程损失:

$$\xi_f = \lambda \frac{l}{d} \frac{1}{2A^2}$$

摩擦系数:

$$\lambda = \begin{cases} 64 / Re & Re < 2300 \\ 0.3164 Re^{-1/4} & 2300 < Re < 10^5 \\ 0.0032 + 0.221 Re^{-0.237} & 10^5 < Re < 10^6 \\ (1.8 \lg Re - 1.5)^{-2} & 10^6 < Re < 3 * 10^6 \end{cases}$$

注意: 完全湍流区 (水力粗糙区) 的显式近似表达式。当雷诺数足够高时, 流动进入完全湍流状态, 此时黏性效应 (与Re 相关) 可忽略, 摩擦因子仅取决于 ϵ/d , ϵ 为绝对粗糙度。

$$\lambda = (1.74 - 2 * \log_{10}(\epsilon / d))^{-2}$$


质量平衡:

$$\dot{m}_{in} = \dot{m}_{out}$$

能量平衡:

$$h_{in} = h_{out}$$

3.模型实现



参数定义

```
import Modelica.Constants;  
outer SysplorerByExample.Demo.PipeTestLib1014.System system;  
extends BasicModel.PartialTwoPorts(dm_flow = 0);  
parameter Modelica.SIunits.Density rho = 1000 "液体密度"  
  annotation (Dialog(tab = "工质参数"));  
parameter Modelica.SIunits.DynamicViscosity mu = 0.0018 "动力粘度"  
  annotation (Dialog(tab = "工质参数"));  
parameter Modelica.SIunits.Length L = 1 "管道长度"  
  annotation (Dialog(tab = "结构参数"));  
parameter Modelica.SIunits.Diameter d(min = 0.0) = 0.02 "管道直径"  
  annotation (Dialog(tab = "结构参数"));  
parameter Modelica.SIunits.Height deltaH = 0 "管道进出口高度差"  
  annotation (Dialog(tab = "结构参数"));  
parameter Modelica.SIunits.Height epsilon = 6e-5 "绝对粗糙度"  
  annotation (Dialog(tab = "结构参数"));  
final parameter Modelica.SIunits.Acceleration g = system.g "重力加速度";
```

相关知识点:

- inner/outer引用
- 类型前缀final
- extends模型重用
- annotation注解

3.模型实现

inner/outer引用

- 组件中引用，通过outer关联

类型前缀final

- 不在参数面板中显示，不能进行修改的值


extends模型重用

- 继承模型

annotation注解 (tab, group)

- 参数面板的设计

3.模型实现



变量定义

```
Modelica.SIunits.Area A = d ^ 2 * Constants.pi / 4 "管道截面积";
Modelica.SIunits.Volume V = L * A "管道容积";
Modelica.SIunits.Diameter Dh = d "水力直径";
Modelica.SIunits.Pressure p_in(start = p_start) "入口压力";
Modelica.SIunits.MassFlowRate m_outflow "出口质量流量";
Modelica.SIunits.ReynoldsNumber Re "管道流动的雷诺数";
Real j = L / A "惯性损失系数";
Real zeta = lambda * L / (2 * d * A ^ 2) "沿程阻力损失系数";
Real lambda "摩擦损失系数";
Modelica.SIunits.SpecificEnthalpy h[2] "从a.b口流入的比焓";
```

相关知识点:

- 内置类型start;
- 数组定义;

3.模型实现

内置类型start:

- 用于设定仿真初始值


数值定义:

- $\text{Real}[3,2] A = \{\{1, 1\}, \{2, 2\}, \{3, 3\}\};$
- $\text{Real } A[3,2] = \{\{1, 1\}, \{2, 2\}, \{3, 3\}\};$

@苏州同元软控信息技术有限公司版权所有，侵权必究

未经许可，不得复印、传播或以其他方式使用

3.模型实现



方程实现

```
equation
//在 Modelica 中，函数名前面的点 (.) 表示完全限定名 (fully qualified name)，用于明确指定函数所在的包
Re = .SysplorerByExample.Demo.PipeTestLib1014.Utilities.Functions.Re(m_outflow, mu, Dh, A);
//雷诺数方程
lambda = .SysplorerByExample.Demo.PipeTestLib1014.Utilities.Functions.lambda(Re, d, epsilon);
//摩擦损失系数方程
m_outflow = -port_b.m_flow;
//出口流量方程
p_in = port_a.p;
//进口压力方程
dp = j * der(m_outflow) + sign(m_outflow) * zeta * m_outflow ^ 2 / rho + rho * g * deltaH;
h[1] = inStream(port_a.h);
h[2] = inStream(port_b.h);
port_b.h = inStream(port_a.h) - system.g*deltaH;
port_a.h = inStream(port_b.h) + system.g*deltaH;
```

相关知识点:

- 函数的定义和调用;
- for循环的使用;
- 数组的使用

3.模型实现

函数定义:

- 雷诺数计算函数;
- 摩擦系数计算函数

相关知识点:

- 函数输入用input关键字声明;
- 函数输出用output关键字声明;
- 函数体是以“algorithm ” 开始的算法区域

@苏州同元软控信息技术有限公司版权所有，侵权必究

3.模型实现

for循环:


- 各分支数量必须一致;
- elseif 可出现0到多次;
- else最多出现一次;

数组使用:


- 数组下标从1开始,下标赋值~~不能为实型~~,必须为参数或常值,不支持动态数组
- 数组的分类:标量、向量、矩阵,向量只有一种表示方式,所以不分行向量和列向量
- 定义数组首先要声明数组的维度

@苏州同元软控信息技术
版权所有,侵权必究

3.模型实现




图标设计



相关知识点:

➤ annotation注解

3.模型实现



说明视图

PipeTestLib > Components > Pipe

PipeTestLib.Components.Pipe

管道

Information


管道压降公式:

$$P_{in} - P_{out} = j \frac{dm_{out}}{dt} + \frac{1}{\rho} \xi_I m_{out}^2 + \rho g \Delta h$$

相关知识点:

➤ annotation注解

3.模型实现



边界条件



```
model BoundaryP "压力边界"
  outer PipeTestLib.System system;
  parameter SI.AbsolutePressure p = 1e5 "压力"
    annotation (Dialog(enable = not use_p_in));
  parameter SI.Temperature T = 293.15 "温度";
  parameter Boolean use_p_in = false "压力由外部接口输入"
    annotation (Dialog(group = "数据来源选项", Evaluate = true, ...));
  Modelica.Blocks.Interfaces.RealInput p_in if use_p_in
    "外部给定压力" annotation (Placement(transformation(...)));
  Interfaces.FluidInterfaces.FluidPort_a port_a
    annotation (Placement(transformation(origin = {100.0, 0.0}, ...)));
equation
  if use_p_in then
    port_a.p = p_in;
  else
    port_a.p = p;
  end if;
  port_a.h = system.cp * (T - 273.15);
```

相关知识点:

- 条件组件
- if判断
- 参数框设计 (复选框)

3.模型实现

构建测试模型：



模型	变量名	变量值
流量源	m_flow	1 kg/s
	T	20 degC
管道1	L	1 m
	d	0.02 m
压力源	p	0.1 MPa

模型	变量名	变量值
压力源	p	0.2 MPa
	T	20 degC
管道1	L	1 m
	d	0.02 m
压力源	p	0.1 MPa

@苏州同元软控信息技术有限公司版权所有，侵权必究

Thanks.

建立知识规范，营造协同生态
积累工业模型，发展可控平台
融入工业创新，共创先进软件

